Under consideration for publication in Theory and Practice of Logic Programming 1

PROGRAMMING PEARL

Computing Convex Hulls with a Linear Solver

FLORENCE BENOY and ANDY KING

Computing Laboratory, University of Kent, UK.

email: {p.m.benoy, a.m.king}@kent.ac.uk

FRED MESNARD

Iremia, Université de La Réunion, France.

email: fred@univ-reunion.fr

Abstract

A programming tactic involving polyhedra is reported that has been widely applied in the polyhedral analysis of (constraint) logic programs. The method enables the computations of convex hulls that are required for polyhedral analysis to be coded with linear constraint solving machinery that is available in many Prolog systems. Keywords: convex hull, polyhedra, abstract interpretation, linear constraints.

1 Introduction

Polyhedra have been widely applied in program analysis (Cousot & Halbwachs, 1978) particularly for reasoning about logic and constraint logic programs. In this context polyhedra have been used in binding-time analysis (Vanhoof & Bruynooghe, 2001), cdr-coded list analysis (Horspool, 1990), argument-size analysis (Benoy & King, 1996), time-complexity analysis (King et al., 1997), high-precision groundness analysis (Codish et al., 2001), type analysis (Sağlam & Gallagher, 1997), termination checking (Codish & Taboch, 1999) and termination inference (Mesnard & Neumerkel, 2001; Genaim & Codish, 2001).

All these techniques use polyhedra to describe relevant properties of the program and manipulate polyhedra using operations that include projection, emptiness checking, inclusion testing for polyhedra, intersection of polyhedra (meet) and the convex hull (join). The classic approach to polyhedral analysis (Cousot & Halbwachs, 1978) uses two representations: (i) frames and rays and (ii) systems of (nonstrict) linear inequalities and employs the Chernikova algorithm to convert between them (Le Verge, 1992). The rationale for this dual representation is that the convex hull can be computed straightforwardly with frames and rays whereas intersection is more simply computed over systems of linear inequalities. A simpler tactic that has been widely adopted in the analysis of logic programs is to use only the linear inequality representation and compute the convex hull by adapting (Benoy &

2 Florence Benoy, Andy King and Fred Mesnard

King, 1996) a relaxation technique proposed in (De Backer & Beringer, 1993). The elegance of this approach is that it enables the convex hull to be computed without recourse to a dual representation: the problem is recast as a projection problem that can be subcontracted to standard linear constraint solving machinery with minimal coding effort. Moreover, the performance is acceptable for many applications. In fact this technique has been widely applied in the analysis of logic programs (Codish & Taboch, 1999; Genaim & Codish, 2001; King et al., 1997; Mesnard & Neumerkel, 2001; Sağlam & Gallagher, 1997). The next section outlines the method and the following section, an example implementation. The final section presents the concluding discussion.

2 Method

Consider two arbitrary polyhedra, P_1 and P_2, represented in standard form: P_1 = {⃗x ∈ Q^n | A_1⃗x ≤ ⃗B_1} P_2 = {⃗x ∈ Q^n | A_2⃗x ≤ ⃗B_2} such that P_1 ≠ ∅ and P_2 ≠ ∅ so that the problem is non-trivial. Note that A_i⃗x ≤ ⃗B_i are non-strict and therefore P_1 and P_2 are both closed. The problem in essence is to compute the smallest polyhedron that includes P_1 and P_2. Interestingly, the convex hull of P_1 ∪ P_2 is not necessarily closed as is illustrated in the following example.

Example 2.1 Consider the 2-dimensional polyhedra P_1 and P_2 defined by:

P_1 = ⃗x ∈ Q^2 | | | | | 1 0 −1 0 0 1 0 −1 ⃗x ≤ 0 0 1 −1 P_2 = ⃗x ∈ Q^2 | | | | 1 −1 −1 1 −1 0 ⃗x ≤ 0 0 0 Observe that P_1 = {⟨0, 1⟩} is a point whereas P_2 = { ⟨x, y⟩ ∈ Q^2 | x = y ∧ 0 ≤ x } is a half-line. Note too that P_1 and P_2 are closed whereas the convex hull of P_1 ∪ P_2 excludes the points { ⟨x, y⟩ ∈ Q^2 | x > 0 ∧ y = x + 1 } and hence is not closed (see the diagram below).

0 1 2 3 x 0 1 2 3 y

P1 and P2

0 1 2 3 x 0 1 2 3 y

convex hull of P1 ∪ P2

0 1 2 3 x 0 1 2 3 y

closure of convex hull of P1 ∪ P2 Since the convex hull of P_1 ∪ P_2 is not necessarily closed, the convex hull cannot always be represented by a system of non-strict linear inequalities; in order to overcome this problem, the closure of the convex hull of P_1 ∪ P_2 is computed. The starting point for our construction is the convex hull of P_1 ∪ P_2 that is given by:

P_H = { ⃗x ∈ Q^n | | | ⃗x = σ_1⃗x_1 + σ_2⃗x_2 ∧ σ_1 + σ_2 = 1 ∧ 0 ≤ σ_1 ∧ A_1⃗x_1 ≤ ⃗B_1 ∧ A_2⃗x_2 ≤ ⃗B_2 ∧ 0 ≤ σ_2 }

Computing Convex Hulls with a Linear Solver 3

To avoid the non-linearity ⃗x = σ_1⃗x_1 +σ_2⃗x_2 the system can be reformulated (relaxed) by putting ⃗y_1 = σ_1⃗x_1 and ⃗y_2 = σ_2⃗x_2 so that ⃗x = ⃗y_1 + ⃗y_2 and A_i⃗y_i ≤ σ_i ⃗B_i to define: P_{CH} = { ⃗x ∈ Q^n | | ⃗x = ⃗y^1 + ⃗y^2 ∧ σ^1 + σ^2 = 1 ∧ 0 ≤ σ^1 ∧ A_1⃗y_1 ≤ σ_1 ⃗B_1 ∧ A_2⃗y_2 ≤ σ_2 ⃗B_2 ∧ 0 ≤ σ_2 }

Observe that P_H ⊆ P_{CH} . Moreover, unlike P_H , P_{CH} is expressed in terms of a system of linear inequalities. Note too that P_{CH} is closed since the projection of a system of non-strict linear inequalities is closed. In fact the following proposition asserts that P_{CH} coincides with the closure of the convex hull of P_1 ∪ P_2.

Proposition 2.1 P_{CH} is the closure of the convex hull of P_1 and P_2.

The proof uses the concept of a recession cone. The recession cone of a polyhedron P , denoted 0^+P , is defined by: 0^+P = {⃗y ∈ Q^n | ∀λ ≥ 0 . ∀⃗x ∈ P . ⃗x + λ⃗y ∈ P }. The intuition is that 0^+P includes a vector ⃗y whenever P includes all the half-lines in the direction of ⃗y that start in P .

Proof Suppose P_i = {⃗x ∈ Q^n | A_i⃗x ≤ ⃗B_i}. Theorem 19.6 of (Rockafellar, 1970) states that the closure of the convex hull of P_1 ∪ P_2 is the set (0^+P_1 + P_2) ∪ (P_1 + 0^+P_2) ∪ (∪{σ_1P_1 + σ_2P_2 | σ_1 + σ_2 = 1 ∧ 0 < σ_1, σ_2}). Intuitively, 0^+P_1 + P_2 is P_2 extended in the directions of half-lines contained within P_1. Let ⃗x ∈ P_i, then ⃗y ∈ 0^+P_i if and only if A_i(⃗x+λ⃗y) ≤ ⃗B_i for all λ ≥ 0 which holds if and only if A_i⃗y ≤ ⃗0 (Rockafellar, 1970)[pp 62]. Therefore 0^+P_1 +P_2 = {⃗x ∈ Q^n | ⃗x = ⃗y_1 +⃗y_2 ∧ A_1⃗y_1 ≤ ⃗0 ∧ A_2⃗y_2 ≤ ⃗B_2} and similarly P_1 + 0^+P_2 = {⃗x ∈ Q^n | ⃗x = ⃗y_1 + ⃗y_2 ∧ A_1⃗y_1 ≤ ⃗B_1 ∧ A_2⃗y_2 ≤ ⃗0}. Furthermore, ∪{σ_1P_1 + σ_2P_2 | σ_1 +σ_2 = 1 ∧ 0 < σ_1, σ_2} = {⃗x ∈ Q^n | σ_1 +σ_2 = 1 ∧ 0 < σ_1, σ_2 ∧ ⃗x = ⃗y_1 + ⃗y_2 ∧ A_1⃗y_1 ≤ σ_1 ⃗B_1 ∧ A_2⃗y_2 ≤ σ_2 ⃗B_2}. Observe that {⃗x ∈ Q^n | ⃗x = ⃗y_1 + ⃗y_2 ∧ A_1⃗y_1 ≤ σ_1 ⃗B_1 ∧ A_2⃗y_2 ≤ σ_2 ⃗B_2} coincides with the sets (i) 0^+P_1 + P_2, (ii) P_1 +0^+P_2 and (iii) ∪{σ_1P_1 + σ_2P_2 | σ_1 +σ_2 = 1 ∧ 0 < σ_1, σ_2} when (i) σ_1 = 0 and σ_2 = 1, (ii) σ_1 = 1 and σ_2 = 0 and (iii) σ_1 +σ_2 = 1 and 0 < σ_1, σ_2 respectively. Therefore P_{CH} is the closure of the convex hull.

This result leads to an algorithm for computing the closure of the convex hull: construct the systems A_i⃗y_i ≤ σ_i ⃗B_i by scaling the constant vectors ⃗B_i by σ_i, add the constraints ⃗x = ⃗y_1 + ⃗y_2, σ_1 + σ_2 = 1 and 0 ≤ σ_i, then eliminate variables other than ⃗x using projection to obtain P_{CH} in terms of ⃗x. Hence the closure of the convex hull can be computed without recourse to another representation. This is illustrated below.

Example 2.2 Returning to example 2.1, consider the systems A_i⃗x ≤ ⃗B_i:

P_1 = { ⟨x, y⟩ ∈ Q^2 | | | x ≤ 0 ∧ −x ≤ 0 ∧ y ≤ 1 ∧ −y ≤ −1 } P_2 = ⟨x, y⟩ ∈ Q^2 | | | | | x − y ≤ 0 ∧ −x + y ≤ 0 ∧ −x ≤ 0

4 Florence Benoy, Andy King and Fred Mesnard

Adding ⃗x = ⃗y_1 + ⃗y_2, σ_1 + σ_2 = 1 and 0 ≤ σ_i leads to the following system:

P_{CH} = ⟨x, y⟩ ∈ Q^2

| | | | | | | x = x_1 + x_2 ∧ y = y_1 + y_2 ∧ σ_1 + σ_2 = 1 ∧ 0 ≤ σ_1 ∧ 0 ≤ σ_2 ∧ x_1 ≤ 0 ∧ −x_1 ≤ 0 ∧ y_1 ≤ σ_1 ∧ −y_1 ≤ −σ_1 ∧ x_2 − y_2 ≤ 0 ∧ −x_2 + y_2 ≤ 0 ∧ −x_2 ≤ 0

Eliminating the variables x_i, y_i and σ_i leads to the solution: P_{CH} = {⟨x, y⟩ ∈ Q^2 | 0 ≤ x ∧ x ≤ y ∧ y ≤ x + 1} Theorem 19.6 of (Rockafellar, 1970), which is used in the proof, asserts that P_{CH} includes P_1 + 0^+P_2 = P_1 + P_2 = {⟨x, y⟩ ∈ Q^2 | x ≥ 0 ∧ y = x + 1} and therefore includes the points {⟨x, y⟩ ∈ Q^2 | x > 0 ∧ y = x + 1}, and hence ensures closure. Note that calculating P_{CH} without the inequalities 0 ≤ σ_1 and 0 ≤ σ_2 – the relaxation advocated in (De Backer & Beringer, 1993) for computing convex hull – gives {⟨x, y⟩ ∈ Q^2 | 0 ≤ x} which is incorrect.

3 Implementation

This section shows how closure of the convex hull can be implemented elegantly using a linear solver in particular the CLP(Q ) library (Holzbaur, 1995). The behaviour of a predicate is described with the aid of modes, that is, + indicates an argument that should be instantiated to a non-variable term when the predicate is called; - indicates an argument that should be uninstantiated; and ? indicates an argument that may or may not be instantiated (Deransart et al., 1996).

3.1 Closed Polyhedra

Closed polyhedra will be represented by lists (conjunctions) of linear constraints of the form c ::= e ≤ e | e = e | e ≥ e where expressions take the form e ::= x | n | n ∗ x | −e | e + e | e − e and n is a rational number and x is a variable. A convenient representation for a closed polyhedron is a (non-ground) list of constraints. This representation is interpreted with respect to a totally ordered (finite) set of variables. The ordering governs the mapping of each variable to its specific dimension. In practise, the ordering on variables is itself represented by the position of each variable within a list. Specifically, if C is a list of linear constraints [c_1, . . . , c_m] and X is a list of variables [x_1, . . . , x_n], then the represented polyhedron is P_{C,X} = {⟨y_1, . . . , y_n⟩ ∈ Q^n | (∧^n_{i=1}x_i = y_i) |=_Q (∧^m_{j=1}c_j )}. Note that although the order of variables in X is significant, the order of the constraints in C is not. Finally, let vars(o) denote the set of variables occurring in the syntactic object o.

Example 3.1 The polyhedron P_1 from example 2.2 can be represented by the lists C_1 = [x = 0, y = 1] and X = [x, y], that is, P_1 = P_{C1,X} . Moreover, P_2 = P_{C2,X} where C_2 = [x = y, x ≥ 0] or alternatively C_2 = [y + z ≥ x, x ≥ y + 2 ∗ z, y ≥ 0, z ≥ 0].

Computing Convex Hulls with a Linear Solver 5

Hence the dimension of P_{C,X} is defined by the length of the list X rather than the number of variables in C.

3.2 Projection

Projection is central to computing the convex hull. The desire, therefore, is to construct a predicate project(+Xs,+Cxs,-ProjectCxs) that is true when for a given list of dimensions Xs and a given list of constraints Cxs, ProjectCxs is the projection of Cxs onto Xs. The specification of such a predicate is given below.

preconditions:

• Xs is a closed list with distinct variables as elements, • Cxs is a closed list of linear constraints, • Cxs is satisfiable.

postconditions:

• Xs is a closed list with distinct variables as elements, • ProjectCxs is a closed list of linear constraints, • vars(ProjectCxs) ⊆ vars(Xs), • P_{Cxs,Xs} = P_{ProjectCxs,Xs}.

Such a predicate can be constructed by adding the given constraints to the store and then invoking the projection facility provided in the CLP(Q ) library, that is, the predicate dump(+Target, -NewVars, -CodedAnswer) (Holzbaur, 1995). Quoting from the manual: “[dump] reflects the constraints on the target variables into a term, where Target and NewVars are lists of variables of equal length and CodedAnswer is the term representation of the projection of constraints onto the target variables where the target variables are replaced by the corresponding variables from

NewVars”. This leads to the following implementation of project:

    :- use_module(library(clpq)).

    project(Xs, Cxs, ProjectCxs) :-
        tell_cs(Cxs),
        dump(Xs, Vs, ProjectCxs), Xs = Vs.

    tell_cs([]).
    tell_cs([C|Cs]) :- {C}, tell_cs(Cs).

Example 3.2 For example, the query project([X, Z], [X < Y, Y < Z], ProjectCs) will correctly bind Cs to [X-Z<0]. However, correctness of this predicate is compromised by existing constraints in the store. For instance, the compound query {X = Z + 1},

project([X, Z], [X < Y, Y < Z], ProjectCs) will fail because constraints

posted within tell cs interact with those already in the store.

To insulate the constraints posted in tell cs, both the variables Xs and the constraints Cxs need to be renamed. Renaming is trivial with the builtin copy term

6 Florence Benoy, Andy King and Fred Mesnard

but care must be taken to ensure that Xs and Cxs are renamed consistently, that is that variable sharing in Xs and Cxs is preserved in the copies. However, in SIC- Stus Prolog copy term(Term, Cpy) copies any constraints in the store that involve variables in Term. For example, the query {X=Y}, copy term(X=Y+1, Cpy) will bind Cpy to A= B+1 where A and B are fresh variables. It will also copy the constraint X = Y by posting the new constraint A = B to the store. To nullify this effect, copy term is called within the scope of call residue. The call

call residue(copy term(X=Y+1, Cpy), Residue) residuates any new constraint

into Residue instead of posting it to the store, thereby copying the term without copying any constraint. Whether residuation is required depends on the particular Prolog system. This leads to the following (SICStus Prolog specific) revision:

project(Xs, Cxs, ProjectCxs) :-
    call_residue(copy_term(Xs-Cxs, CpyXs-CpyCxs), _),
    tell_cs(CpyCxs),
    dump(CpyXs, Vs, ProjectCxs), Xs = Vs.

Example 3.3 Using this revision, the query {X = Z + 1}, project([X, Z], [X < Y, Y < Z],

ProjectCs) will succeed binding ProjectCs to [X-Z<0]. However, adding Z = 5 to

the list of constraints induces an error. The problem is that posting the constraints binds Z to 5 so that dump is called with its first argument instantiated to a list that contains a non-variable term.

A pre-processing predicate prepare dump is therefore introduced to ensure that

dump is called correctly. The following revision to project, in effect, extends the

facility provided by dump to capture constraints over both uninstantiated and instantiated variables:

project(Xs, Cxs, ProjectCxs) :-
    call_residue(copy_term(Xs-Cxs, CpyXs-CpyCxs), _),
    tell_cs(CpyCxs),
    prepare_dump(CpyXs, Xs, Zs, DumpCxs, ProjectCxs),
    dump(Zs, Vs, DumpCxs), Xs = Vs.

prepare_dump([], [], [], Cs, Cs).
prepare_dump([X|Xs], YsIn, ZsOut, CsIn, CsOut) :-
    (ground(X) ->
        YsIn = [Y|Ys],
        ZsOut = [_|Zs],
        CsOut = [Y=X|Cs]
    ;
        YsIn = [_|Ys],
        ZsOut = [X|Zs],
        CsOut = Cs
    ),
    prepare_dump(Xs, Ys, Zs, CsIn, Cs).

Computing Convex Hulls with a Linear Solver 7

The literal prepare dump(+Xs, +Ys, -Zs, ?CsIn, -CsOut) is true for a given list

Xs which contains either variables or numbers (or a mixture of the two) and a given

list Ys which contains only variables, if

• Zs is the list obtained by substituting the non-variable terms of Xs with fresh variables and • CsOut is an open ended list of equality constraints with CsIn at its end that contains one equality constraint for each number in Xs. Each constraint equates a numeric element of Xs with the element of Ys that is in the same list position.

The call prepare dump([X1, 1, X3, 2], [A, B, C, D], Zs, CsIn, CsOut), for instance, will bind Zs to [X1, A,X3, B] and CsOut to [B=1,D=2|CsIn]. The predicate ensures that dump is called with its first argument bound to a list of free variables even when the list Xs includes numbers. In the CLP(Q ) library, numbers coincide with rationals which are represented as compound (ground) terms of the form rat(n, d) where n and d are integers. The ground(X) test effectively checks whether X is instantiated to a number; the test number(X) is inappropriate since it would always fail.

Example 3.4 Consider again example 3.1. The second representation of P_2 can be simplified by using projection as follows:

| ?- Cs = [Y+Z>=X,X>=Y+2*Z,Y>=0,Z>=0], project([X,Y], Cs, ProjectCs).
ProjectCs = [Y>=0,X=Y] ? ;
no

The system Cs is expressed over 3 variables and therefore defines a 3 dimensional space. Intuitively, the projection onto [X, Y] is the shadow cast by P_{Cs,[X,Y,Z]} onto the 2 dimensional space over X and Y. The projection ProjectCs in fact defines a half-line confined to the first quadrant since, by rearranging Cs, it follows that P_{Cs,[X,Y,Z]} = {⟨x, y, z⟩ ∈ Q^3 | x = y ∧ 0 ≤ y ∧ z = 0}.

3.3 Convex Hull

The specification for the main predicate convex hull(+Xs, +Cxs, +Ys, +Cys,

-Zs, -Czs), and then its code, is given below.

preconditions:

• Xs is a closed list with distinct variables as elements and likewise for Ys, • Xs and Ys have the same length, • vars(Xs) ∩ vars(Ys) = ∅, • Cxs and Cys are closed lists of linear constraints, • Cxs and Cys are both satisfiable, • vars(Cxs) ⊆ vars(Xs) and vars(Cys) ⊆ vars(Ys).

8 Florence Benoy, Andy King and Fred Mesnard

postconditions:

• Xs, Ys and Zs are closed lists with distinct variables as elements, • Zs is the same length as both Xs and Ys, • Czs is a closed list of linear constraints, • vars(Czs) ⊆ vars(Zs) and (vars(Xs) ∪ vars(Ys)) ∩ vars(Zs) = ∅, • P_{Czs,Zs} is the closure of the convex hull of P_{Cxs,Xs} ∪ P_{Cys,Ys}.

convex_hull(Xs, Cxs, Ys, Cys, Zs, Czs) :-
    scale(Cxs, Sig1, [], C1s),
    scale(Cys, Sig2, C1s, C2s),
    add_vect(Xs, Ys, Zs, C2s, C3s),
    project(Zs, [Sig1 >= 0, Sig2 >= 0, Sig1+Sig2 = 1|C3s], Czs).

scale([], _, Cs, Cs).
scale([C1|C1s], Sig, C2s, C3s) :-
    C1 =.. [RelOp, A1, B1],
    C2 =.. [RelOp, A2, B2],
    mul_exp(A1, Sig, A2),
    mul_exp(B1, Sig, B2),
    scale(C1s, Sig, [C2|C2s], C3s).

mul_exp(E1, Sigma, E2) :- once(mulexp(E1, Sigma, E2)).

mulexp( X, _, X) :- var(X).
mulexp(N*X, _, N*X) :- ground(N), var(X).
mulexp( -X, Sig, -Y) :- mulexp(X, Sig, Y).
mulexp(A+B, Sig, C+D) :- mulexp(A, Sig, C), mulexp(B, Sig, D).
mulexp(A-B, Sig, C-D) :- mulexp(A, Sig, C), mulexp(B, Sig, D).
mulexp( N, Sig, N*Sig) :- ground(N).

add_vect([], [], [], Cs, Cs).
add_vect([U|Us], [V|Vs], [W|Ws], C1s, C2s) :-
    add_vect(Us, Vs, Ws, [W = U+V|C1s], C2s).

The predicate mulexp(?E1, ?Sigma, -E2) scales the numeric constants that occur within E1 by the variable Sigma, providing they are not coefficients of variables, to obtain the expression E2. Note that Sigma is a variable and the expression E1 may be a variable, hence both E1 and Sigma have mode ? rather than +. Since a nonground representation is employed for expressions, the test var(X) is used to determine whether the expression is a variable. As before, the test ground(N) detects numeric constants – rational numbers – which are the only type of subexpressions that are ground. Observe that mulexp can return more than one solution, for example,

mulexp(X, Sig, E2) generates E2 = X; X = -( A), E2 = -( A); X = -(-( A)),
E2 = -(-( A)) etc as solutions. Thus the pruning operator once is applied within
mul exp(?E1, ?Sigma, -E2) to prevent erroneous solutions.

Computing Convex Hulls with a Linear Solver 9

The predicate scale(+C1s, ?Sigma, ?C2s, -C3s) scales each constraint within the list C1s by the variable Sigma. Each constraint consists of a binary operator and two expressions, and scaling is applied to the numeric constants in each expression as specified by mul exp. For example, scale([X+2 >= 1+Y, Y = Z], Sigma, Tail,

ScaledCs) binds ScaledCs to [Y = Z, X+2*Sigma >= 1*Sigma+Y | Tail]. Note

that scale finesses the problem of putting Cxs and Cys into the standard form A_i⃗y_i ≤ ⃗B_i before applying scaling. In standard form, X+2 >= 1+Y is Y-X =< 1 but scaling constants on both sides of the relational operator preserves equivalence in that X+2*Sig >= 1*Sig+Y is equivalent to Y-X =< 1*Sig. The use of a difference list avoids an unnecessary call to append in the body of convex hull.

The predicate add vect(+Us, +Vs, -Ws, ?C1s, -C2s) operates on the lists Us

= [U_1, . . ., U_n] and Vs = [V_1, . . ., V_n] which correspond to the vectors ⃗y_1 and

⃗y_2 (as introduced in section 2). The argument Ws is instantiated to another list of variables [W_1, . . ., W_n], which corresponds with ⃗x. The predicate creates the system of equalities [W_1 = U_1+V_1, . . ., W_n = U_n+V_n] corresponding to the system ⃗x = ⃗y_1 + ⃗y_2. The scaled constraints output by the two calls to scale are passed to

add vect via its accumulator and thereby combined with the system of equalities.

For example, the call add vect([X1,Y1], [X2, Y2], Ws, Tail, Cs) returns the bindings Cs = [ A=Y1+Y2, B=X1+X2|Tail] and Ws = [ B, A].

The predicate convex hull(Xs, Cxs, Ys, Cys, Zs, Czs) takes, as input, two lists of constraints (Cxs and Cys) and their corresponding lists of variables (Xs and Ys) and produces as output a single list of constraints Czs over the variables Zs that represents the closure of the convex hull of the two input polyhedra. If Xs and Ys are not variable disjoint, then the pre-requisite can be satisfied by appropriately renaming variables. Specifically, the variables Xs and constraints Cxs can be renamed with copy term(Xs-Cxs, CpyXs-CpyCxs) and the call

convex hull(Xs, Cxs, Ys, Cys, Zs, Czs) replaced with convex hull(CpyXs,
CpyCxs, Ys, Cys, Zs, Czs). Since the integrity of the constraint store is pre-

served by project and since project is the only source of interaction with the store, then it follows that convex hull also does not side-effect any existing constraints. The following is an illustrative example.

Example 3.5 Running this code on the data of Example 2.2 gives:

| ?- convex_hull([X1,Y1],[X1=0,Y1=1],[X2,Y2],[X2>=0,Y2=X2],V,S).
S = [_A>=0,_A-_B>=-1,_A-_B=<0],
V = [_A,_B] ? ;
no

4 Discussion

This section discusses the method proposed in the paper, comparing it with related techniques. The Chernikova method is exponential in the worst-case (Le Verge, 1992) and the Fourier-Motzkin method, like all projection techniques over linear

10 Florence Benoy, Andy King and Fred Mesnard

P1 P2 -1 1 x

-1

1 y

Q_1 Q2

-1

1 x

-1

1 y

-1

1 z

-1 1 x

-1

1 y

-1

1 x

-1

1 y

-1

1 z

Fig. 1. (i) P1 and P2, (ii) Q1 and Q2, (iii) conv(P1 ∪ P2), (iv) conv(Q1 ∪ Q2)

inequalities (Chandru et al., 2000), is also exponential. The exponential behaviour of both methods stems from the same source: the possibly exponential relationship between the number of vertices and the number of half-spaces that define a polyhedron. In fact the problem of calculating the closure of the convex hull of two polyhedra is also exponential even for bounded polyhedra (polytopes). This can be demonstrated by considering the so-called cross polytope in n-dimensions which is the polyhedron with the vertex set {⟨±1, 0, . . . , 0⟩, ⟨0, ±1, . . . , 0⟩, . . . , ⟨0, 0, . . . , ±1⟩}. The cross polytope can be defined by no less than 2^n inequalities yet can arise as the convex hull of two polyhedra both of which can be defined with O(n) inequalities. Specifically consider the n-dimensional polyhedra P_1 = {⟨x_1, . . . , x_n⟩ ∈ Q^n | ( ∑_n i=1 −x_i ≤ 1) ∧ (∧^n_{j=1}x_j ≤ 0)} P_2 = {⟨x_1, . . . , x_n⟩ ∈ Q^n | ( ∑_n i=1 x_i ≤ 1) ∧ (∧^n_{j=1} − x_j ≤ 0)}

Because P_1 and P_2 are polytopes, they can be expressed in terms of their vertices: P_1 = conv({⟨0, 0, . . . , 0⟩, ⟨−1, 0, . . . , 0⟩, ⟨0, −1, . . . , 0⟩, . . . , ⟨0, 0, . . . , −1⟩}) P_2 = conv({⟨0, 0, . . . , 0⟩, ⟨1, 0, . . . , 0⟩, ⟨0, 1, . . . , 0⟩, . . . , ⟨0, 0, . . . , 1⟩}) Since ⟨0, 0, . . . , 0⟩ is convexly spanned by ⟨1, 0, . . . , 0⟩ and ⟨−1, 0, . . . , 0⟩, it follows that cl(conv(P_1 ∪ P_2)) = conv(P_1 ∪ P_2) = conv({⟨±1, 0, . . . , 0⟩, ⟨0, ±1, . . . , 0⟩, . . ., ⟨0, 0, . . . , ±1⟩}) which is the n-dimensional cross polytope. The 2 and 3 dimensional cases are denoted in Figure 1 by (i) P_1 and P_2 and (ii) Q_1 and Q_2 respectively for which the cross polytopes are a solid square and an octahedron. Hence the problem of calculating the closure of the convex hull is intrinsically exponential irrespective of the algorithm employed.

Computing Convex Hulls with a Linear Solver 11

Example 4.1 The following query illustrates how the hull algorithm yields an exponential number of inequalities for the 4 dimensional case.

| ?- Xs = [X1, X2, X3, X4], Ys = [Y1, Y2, Y3, Y4],
     Cxs = [-1 =< X1+X2+X3+X4, X1 =< 0, X2 =< 0, X3 =< 0, X4 =< 0],
     Cys = [ Y1+Y2+Y3+Y4 =< 1, 0 =< Y1, 0 =< Y2, 0 =< Y3, 0 =< Y4],
     convex_hull(Xs, Cxs, Ys, Cys, Zs, Czs),
     Zs = [A, B, C, D].

Czs = [A-B+C+D>=-1, A+B-C-D=<1, A+B+C+D>=-1, A-B-C-D=<1,
       A-B-C+D>=-1, A+B+C-D=<1, A+B-C+D>=-1, A-B+C-D=<1,
       A-B+C-D>=-1, A+B-C+D=<1, A+B+C-D>=-1, A-B-C+D=<1,
       A-B-C-D>=-1, A+B+C+D=<1, A+B-C-D>=-1, A-B+C+D=<1] ? ;

no

However, it would be wrong to conclude from these examples that the frame and ray representation is preferable – inequalities are unavoidable since they are required for other polyhedral operations.

Despite the scaling problems that are inherent to any convex hull algorithm, in practise the technique proposed in this paper has been widely applied in logic programming (Codish & Taboch, 1999; Genaim & Codish, 2001; King et al., 1997; Mesnard & Neumerkel, 2001; Sağlam & Gallagher, 1997), mostly to satisfaction. For example, in the context of inferring termination conditions for logic programs this method is feasible since it accounts for 42% of this first pass of the analysis and the first pass itself constitutes only 23% of the total analysis time (Mesnard & Neumerkel, 2001). Whether the approach presented in this paper is applicable depends on the application context. When only standard domain operations are required and performance is not an issue, this method has much to commend it. However, when the application has to additionally reason, say, about integral points (Ancourt, 1991; Quinton et al., 1997) or parameterised polyhedra (Loechner & Wilde, 1997) then specialised polyhedral libraries are required. Further, if performance is important, then recourse should be made to a polyhedral library, since a state-of-the-art implementation employing the Chernikova algorithm (Bagnara et al., 2002), will outperform the approach presented here.

We have presented a Prolog program for computing convex hulls using linear solver machinery. As Holzbaur’s library is also available for CIAO Prolog, ECLiPSe, XSB and Yap Prolog, the technique can be easily adapted to these systems. The method is a reasonable compromise between conciseness, clarity and efficiency and variants of this program have now been widely deployed.

Acknowledgements Thanks are due to Mats Carlsson, Bart Demoen, Pat Hill, Joachim Schimpf and Raimund Seidel and, of course, the anonymous referees.

12 Florence Benoy, Andy King and Fred Mesnard

References

Ancourt, C. 1991 (March). Gén’eration Automatique de Code de Transfert pour Multiprocesseurs à Mémoires Locales. Ph.D. thesis, Université Paris 6.

Bagnara, R., Ricci, E., Zaffanella, E., & Hill, P. M. (2002). Possibly Not Closed Convex Polyhedra and the Parma Polyhedra Library. Pages 213–229 of: Hermenegildo, M. V., & Puebla, G. (eds), Static Analysis Symposium. Lecture Notes in Computer Science, vol. 2477. Springer-Verlag. See also http://www.cs.unipr.it/ppl/.

Benoy, F., & King, A. (1996). Inferring Argument Size Relationships with CLP(R). Pages 204–223 of: Gallagher, J. P. (ed), Logic-based Program Synthesis and Transformation (Selected Papers). Lecture Notes in Computer Science, vol. 1207. Springer-Verlag.

Chandru, V., Lassez, C., & Lassez, J.-L. 2000 (March). Qualitative Theorem Proving in Linear Constraints. International Symposium on Artificial Intelligence and Mathematics. Long version to appear in the Annals of Mathematics and Artificial Intelligence.

Codish, M., & Taboch, C. (1999). A Semantic Basis for the Termination Analysis of Logic Programs. The Journal of Logic Programming, 41(1), 103–123.

Codish, M., Genaim, S., Sondergaard, H., & Stuckey, P. (2001). Higher-Precision Groundness Analysis. Pages 135–149 of: Codognet, P. (ed), International Conference on Logic Programming. Lecture Notes in Computer Science, vol. 2237. Springer-Verlag.

Cousot, P., & Halbwachs, N. (1978). Automatic Discovery of Linear Restraints among Variables of a Program. Pages 84–97 of: Principles of Programming Languages. ACM Press.

De Backer, B., & Beringer, H. (1993). A CLP language handling disjunctions of linear constraints. Pages 550–563 of: Warren, D. S. (ed), International Conference on Logic Programming. MIT Press.

Deransart, P., Ed-Dbali, A., & Cervoni, L. (1996). Prolog: The Standard. Springer-Verlag. Genaim, S., & Codish, M. (2001). Inferring Termination Conditions for Logic Programs using Backwards Analysis. Pages 681–690 of: Nieuwenhuis, R., & Voronkov, A. (eds), International Conference on Logic for Programming, Artificial Intelligence and Reasoning. Lecture Notes in Artificial Intelligence, vol. 2250. Springer-Verlag.

Holzbaur, C. (1995). OFAI clp(Q,R) Manual. Tech. rept. TR-95-09. Austrian Research Institute for Artificial Intelligence (ÖFAI), Schottengasse 3, A-1010 Vienna, Austria.

Horspool, R. N. (1990). Analyzing List Usage in Prolog. University of Victoria. King, A., Shen, K., & Benoy, F. (1997). Lower-bound Time-Complexity Analysis of Logic Programs. Pages 261–276 of: Maluszynski, J. (ed), International Symposium on Logic Programming. MIT Press.

Le Verge, H. (1992). A note on Chernikova’s algorithm. Tech. rept. 635. IRISA, Campus Universitaire de Beaulieu, Rennes, France.

Loechner, V., & Wilde, D. K. (1997). Parameterized Polyhedra and their Vertices. International Journal of Parallel Programming, 25(6), 525–549. See also http://icps.ustrasbg.fr/polylib/.

Mesnard, F., & Neumerkel, U. (2001). Applying Static Analysis Techniques for Inferring Termination Conditions of Logic Programs. Pages 93–110 of: Cousot, P. (ed), Static Analysis Symposium. Lecture Notes in Computer Science, vol. 2126. Springer-Verlag.

Quinton, P., Rajopadhye, S. V., & Risset, T. (1997). On Manipulating Z-Polyhedra using a Canonical Representation. Parallel Processing Letters, 7(2), 181–194.

Rockafellar, R. T. (1970). Convex Analysis. Princeton University Press. Sağlam, H., & Gallagher, J. P. (1997). Constrained Regular Approximation of Logic Programs. Pages 282–299 of: Fuchs, N. E. (ed), Logic Programming Synthesis and Transformation (Selected Papers). Springer-Verlag.

Computing Convex Hulls with a Linear Solver 13

Vanhoof, W., & Bruynooghe, M. (2001). Binding-Time Annotations Without Binding- Time Analysis. Pages 707–722 of: Nieuwenhuis, R., & Voronkov, A. (eds), International Conference on Logic for Programming, Artificial Intelligence and Reasoning. Lecture Notes in Artificial Intelligence, vol. 2250. Springer-Verlag.